

PRÁCTICA 6. CÁLCULO INTEGRAL EN VARIAS VARIABLES. ECUACIONES DIFERENCIALES.

1 INTRODUCCIÓN.

El alumno debe ir realizando todos los EJEMPLOS que aparecen en esta práctica e ir anotando los resultados en la hoja que se le ha entregado. Además de estos ejemplos, hay planteados unos EJERCICIOS, que, a diferencia de lo que ocurre con los EJEMPLOS, no vienen desarrollados en el lenguaje Maxima, sino que tendrá que ser el alumno el que tenga que introducirlos para obtener el resultado.

2 INTEGRALES MÚLTIPLES.

2.1 Integrales dobles

En una práctica anterior hemos visto cómo podemos utilizar Maxima para resolver integrales de funciones de una variable. ¿Qué necesitamos para calcular integrales de funciones de varias variables? Básicamente, el teorema de Fubini, que nos permite calcular una integral doble como dos integrales simples (idem para integrales triples).

```
--> /*EJEMPLO 1: La integral de la función f(x,y)=3xy en [0,1]x[2,6]*/
      integrate(integrate(3*x*y,x,0,1),y,2,6);
```

```
--> /*EJEMPLO 2: Cambiando el orden de integración*/
      integrate(integrate(3*x*y,y,2,6),x,0,1);
```

```
--> /*EJEMPLO 3: La integral de la función f(x,y)=x^2+y^2 en
      [0,2]x[1,5]*/
      integrate(integrate(x^2+y^2,x,0,2),y,1,5);
```

Podemos calcular estas integrales en recintos más complicados:
EJEMPLO 4: Si queremos calcular la integral de $f(x,y)=x$ en el primer cuadrante del círculo $x^2+y^2 \leq 1$, es aconsejable que en primer lugar dibujemos el conjunto:

```
--> /*EJEMPLO 4:*/
      load(draw)$
      draw2d(implicit(x^2+y^2=1,x,0,1,y,0,1));
```

Así podemos ver la variabilidad de cada variable (en este caso, si x varia en $[0,1]$, se tiene que y varia entre 0 y $(1-x^2)^{(1/2)}$, por lo que podemos calcular la integral sin más que

```
--> /*EJEMPLO 4 (cont.):*/  
  
      integrate(integrate(x,y,0,sqrt(1-x^2)),x,0,1);
```

Como nos da un error (ya que inicialmente no hemos indicado que la variable x toma un valor entre 0 y 1 , pondremos

```
--> /*EJEMPLO 4 (cont.):*/  
      integrate(integrate(x,y,0,sqrt(abs(1-x^2))),x,0,1);
```

Observemos que que en este ejemplo no ha sido necesario cambiar a coordenadas polares. Para integrales en varias variables, interesa cambiar de variable cuando el dominio o la función son complicados. En el primer caso, si no podemos escribir el dominio de una de las formas habituales (con x variando entre constantes e y variando entre funciones de x ; o al revés) tendremos que cambiar de variable nosotros (Maxima no lo hace automáticamente). El caso de funciones complicadas no nos debe preocupar, pues Maxima se encarga de resolver la integral en la inmensa mayoría de los casos, y si no se puede resolver directamente, siempre queda la posibilidad de aproximar el valor de integral (lo que haremos usando `quad_quags(f(x),x,a,b)` o `romberg(f(x),x,a,b)`; ambas sentencias nos dan el valor aproximado de la integral de $f(x)$ en $[a,b]$).

```
--> /*EJEMPLO 5: Calcular la integral de la función f(x,y)=exp(x/y) ,  
      en el conjunto de puntos (x,y) tal que 0<=y^3<=x<=y^2*/  
  
      /*Primero calculamos los puntos de corte y dibujamos la gráfica  
      de la región*/  
  
      solve(y^3=y^2,y);  
      draw2d(color=blue,implicit(y^3=x,x,0,1,y,0,1),color=red,  
      implicit(y^2=x,x,0,1,y,0,1));
```

```
--> /*EJEMPLO 5 (cont.): De lo hecho se deduce que la integral que  
      queremos calcular es:*/  
  
      integrate(integrate(exp(x/y),x,y^3,y^2),y,0,1);
```

Veamos un ejemplo con función a integrar relativamente sencilla pero con dominio no tanto. Aquí observaremos que no hay manera fácil de escribir la integral en términos de x e y . Parece mejor cambiar a coordenadas polares. Aunque antes de hacerlo podemos echar un vistazo al dominio. Recordamos que el rango de valores de las variables x e y es algo que debemos decidir nosotros:

```
--> /*EJEMPLO 6: Calcular la integral de la función  $f(x,y)=x^2+y^2$  ,
      en el conjunto de puntos  $(x,y)$  tal que  $(x^2+y^2)^2 \leq 4(x^2-y^2)$ ,
      con  $x \geq 0$ :*/

      load(draw)$
      draw2d(color=blue,implicit((x^2+y^2)^2=4*(x^2-y^2),
      x,0,2,y,-1,1));
```

Operando aparte y viendo la gráfica, llegamos a la conclusión que el argumento t varía en $[-\pi/4, \pi/4]$, mientras que el módulo r varía entre 0 y $2 \cdot (\cos(2t))^{1/2}$. Además, la función $f(x,y)$ se transforma en $f(r,t)=r^2$. Por tanto, y sin que se os olvide multiplicar f por el jacobiano (que es r), tendremos

```
--> /*EJEMPLO 6 (cont.): La integral a calcular en polares es:*/

      integrate(integrate(r^3,r,0,2*sqrt(cos(2*t))),t,-%pi/4,%pi/4);
```

Como de nuevo nos da un error (por la raíz cuadrada), pondremos:

```
--> /*EJEMPLO 6 (cont.): La integral a calcular en polares es:*/

      integrate(integrate(r^3,r,0,2*sqrt(abs(cos(2*t)))),t,-%pi/4,%pi/4);
```

EJERCICIO 1: Calcular la integral de las siguientes funciones en los recintos determinados por las siguientes ecuaciones:

- a) $f(x,y)=(4x^2-y^2)^{1/2}$, $x=1$, $y=0$, $y=x$.
- b) $f(x,y)=x \cdot \exp(-x^2/y)$, $x=0$, $y=1$, $y=x^2$.
- c) $f(x,y)=y$, $y \geq 0$, $x^2+y^2=a^2$, $y^2=2ax$, $x=0$.

EJERCICIO 2: Calcular el volumen limitado superiormente por el cono $4x^2+4y^2-z^2=0$, inferiormente por el plano $z=0$ y lateralmente por el cilindro $x^2+(y-2)^2=4$.

2.2 Integrales triples

Análogamente a las integrales dobles, podemos calcular integrales triples. Veamos en el ejemplo siguiente como se puede obtener el volumen de un elipsoide (parecido a un balón de rugby) de semiejes a , b y c . Este ejercicio lo vamos a realizar mediante integrales triples (recordamos que el volumen, mediante integrales triples, se obtiene integrando $dx dy dz$ en el volumen que se quiere calcular:

```
--> /*EJEMPLO N° 6 BIS: Calculamos el volumen de un elipsoide
con semiejes a=4,b=2 y c=2*/
/*Lo hacemos en el primer octante por comodidad de los
límites de integración*/

zmax:2*sqrt(1-x^2/16-y^2/4)$
ymax:2*sqrt(1-x^2/16)$
integrate(integrate(integrate(1,z,0,zmax),y,0,ymax),x,0,4);
```

Notemos como nos da un aviso sobre que el límite de integración ha de ser real. Por ello, pondremos

```
--> assume(1-x^2/16-y^2/4>0)$
assume(1-x^2/16>0)$
zmax:2*sqrt(1-x^2/16-y^2/4)$
ymax:2*sqrt(1-x^2/16)$
integrate(integrate(integrate(1,z,0,zmax),y,0,ymax),x,0,4);
```

Y como esto es solo el volumen en el 1er octante, habremos de multiplicar por 8 para obtener el volumen total:

```
--> %*8;
```

```
--> /*EJEMPLO N° 7: Usando integración triple, calcular el volumen
comprendido entre  $x^2+y^2 \leq z^2$  (cono) y  $x^2+y^2+z^2 \leq 4$  (esfera)
representando previamente la región*/

/*Realizamos la representación gráfica del recinto*/
remvalue(all)$
load(draw)$
draw3d([color=red,implicit(x^2+y^2+z^2=4,x,-5,5,y,-5,5,z,-5,5),
color=blue,implicit(x^2+y^2=z^2,x,-5,5,y,-5,5,z,-5,5)]);
```

Para conocer el recinto de integración, debemos hallar los puntos donde se cortan la esfera y el cono. En este caso es trivial, ya que si sustituimos una ecuación en la otra, obtendremos como intersección la circunferencia $x^2+y^2=2$. Ésta será por tanto la proyección de dicha intersección sobre el plano XY. Por tanto, el volumen pedido será el doble (ya que la figura tiene intersección en los 8 octantes y nosotros vamos a calcular solo el de la parte superior) de la integral triple de $dx dy dz$, donde x e y varían en la circunferencia anterior, mientras que z varía entre el cono y la esfera

```
--> 2*integrate(integrate(integrate(1,z,(x^2+y^2)^(1/2),
(4-x^2+y^2)^(1/2)),
y,-(2-x^2)^(1/2),(2-x^2)^(1/2)),x,-sqrt(2),sqrt(2));
```

Si ejecutamos la sentencia anterior, notamos que Maxima no nos da el resultado. Por tanto, haremos un cambio a coordenadas cilíndricas, siendo la integral a resolver la dada por (la función será r , que es el jacobiano, mientras que z varía entre el cono (que tiene por ecuación en cilíndricas: r) y la esfera (con ecuación en cilíndricas: $(4-r^2)^{1/2}$); r varía entre 0 y $\sqrt{2}$, mientras que θ varía entre 0 y 2π):

```
--> 2*integrate(integrate(integrate(r,z,r,(4-r^2)^(1/2)),
    r,0,sqrt(2)),t,0,2*pi);
```

Como también tenemos problemas, realizaremos nosotros aparte la primera de las integrales (resultará $r((4-r^2)^{1/2}-r)$ y ahora obtendremos la integral doble restante

```
--> 2*(integrate(r*((4-r^2)^(1/2)-r),r,0,sqrt(2)),t,0,2*pi);
```

3 ECUACIONES DIFERENCIALES.

En esta última parte de la práctica 6, vamos a ver como podemos resolver con Maxima los ejemplos más típicos de ecuaciones diferenciales ordinarias (edo) especialmente las de primer orden, y algunas de 2do orden.

3.1 Introducción a las EDO

En primer lugar veremos como podemos comprobar que una función determinada es solución explícita de una EDO. Para ello, solo tenemos que sustituir la función en la ec. diferencial, usando el comando `ratsim()` como vemos a continuación:

```
--> /*EJEMPLO N° 7 BIS: Probamos que la función f(x)=2/(2+x)
    es solución de la edo
    xy'+y-y^2=0*/
```

```
y(x):=2/(2+x)$
ratsimp(x*diff(y(x),x)+y(x)-y(x)^2);
```

```
--> /*EJEMPLO N° 8: Comprobamos que la familia uniparamétrica de
    funciones y=1+c*e^(-x), es solución de la edo de primer orden
    y'+y=1*/
```

```
kill(y)$
y(x):=1+c*%e^(-x)$
is(ratsimp(diff(y(x),x)+y(x)=1));
```

Para representar gráficamente algunas soluciones particulares, podemos hacer una lista dando algunos valores a la constante c. Por ejemplo, generamos una lista que llamamos soluciones con valores de $c = -3, -2, 0, 1, 2, 3$:

```
--> /*EJEMPLO N° 8 (cont.):*/

soluciones:makelist(1+c*%e^(-x),c,-3,3);
```

Y podemos representarlas en un sólo gráfico, con la opción `plot2d`:

```
--> /*EJEMPLO N° 8 (cont.):*/

plot2d(soluciones,[x,-3,3]);
```

```
--> /*EJEMPLO N° 9: Comprobamos que  $y(t)=c_1\cos(4t)+c_2\sin(4t)$ 
es solución explícita de la edo de segundo orden
 $y''+16y=0$ */

kill(y)$
y(t):=c1*cos(4*t)+c2*sin(4*t)$
is(ratsimp(diff(y(t),t,2)+16*y(t))=0);
```

También podemos hacer lo mismo en el caso en que la solución venga dada en forma implícita:

```
--> /*EJEMPLO N° 10: La ecuación  $x^2+y^2=25$ , siendo y función
implícita de x, es solución implícita de la ecuación
 $y*y'+x=0$ */

kill(y,x)$
depends(y,x)$
diff(x^2+y^2=25,x)$
solve(%,diff(y,x));
```

```
--> /*EJEMPLO N° 10 (cont):*/

ec:y*'diff(y,x)+x=0$
solve(ec,diff(y,x));
```

```
--> /*EJEMPLO N° 11: Comprobamos que  $e^{(x*y)}+y=x-1$  es solución
implícita de la ecuación
 $y'=(\exp(-x*y)-y)/(\exp(-x*y)+x)$ */

kill(x,y)$
depends(y,x)$
ec:%e^(x*y)+y=x-1$
diff(ec,x);
solve(%,diff(y,x));
ec2:-(y*e^(x*y)-1)/(x*e^(x*y)+1)=
(%e^(-x*y)-y)/(%e^(-x*y)+x);
ratsimp(%);
```

También podemos comprobar que se cumple un PVI (problema de Cauchy): Por ejemplo, el PVI dado por $y'=(3/2)*y^{(2/3)}$ $y(0)=1$ tiene una única solución, puesto que $\partial f/\partial y$ no se anula en el punto $(x,y)=(0,1)$. Veamos que la función $y=((x+2)^3)/8$ es la única solución de este PVI:

```
--> /*EJEMPLO N° 12:*/

kill(all)$
y(x):=(x+2)^3/8$
is(diff(y(x),x)=3/2*y(x)^(2/3));

--> /*EJEMPLO N° 12 (cont): Y además verifica la
condición inicial*/

is(y(0)=1);
```

```
--> /*EJEMPLO N° 13: Comprobamos que la función
      y=2^(1/2)*sin(2*x)-1/2 es solución del PVI
      y''+4y=-2, y(pi/8)=1/2, y'(pi/8)=2*/
```

```
kill(all)$
y(x):=sqrt(2)*sin(2*x)-1/2$
is(ratsimp(diff(y(x),x,2)+4*y(x))=-2);
```

```
--> /*EJEMPLO N° 13 (cont): Y además verifica las
      condiciones iniciales*/
```

```
is(y(%pi/8)=1/2);
diff(y(x),x)$
%,x:%pi/8;
```

3.2 EDO en variables separadas (integración directa)

Maxima se puede utilizar no sólo para resolver ciertas EDO's, sino también para aplicar los métodos vistos en clase y así comprobar la resolución paso a paso. Por ejemplo, si queremos resolver la ecuación $y' = \log(x) + 1 - x^2 + (x+2)^{1/2}$ estamos en el caso más sencillo, pues se puede resolver por integración directa:

```
--> /*EJEMPLO N° 14:*/
```

```
integrate(log(x)+1-x^2+sqrt(x+2),x);
```

Notemos que la constante de integración no parece (así que cuando sea necesaria habrá que incorporarla a mano).

Maxima tiene varias órdenes para la resolución directa de EDO's, una de ellas es "ode2". Si resuelves la ecuación anterior con ode2 observarás que en este caso no hay, esencialmente, ninguna diferencia:

```
--> /*EJEMPLO N° 15:*/
```

```
ode2('diff(y,x)=log(x)+1-x^2+sqrt(x+2),y,x);
```

En este caso sí que aparece el parámetro; si te dan alguna condición inicial podemos calcularlo mediante la orden "ic1": Por ejemplo, si en el caso anterior exigimos que $y(1)=1$:

```
--> ic1(%,x=1,y=1);
```

Supongamos ahora que queremos resolver la EDO $x*y'+y=1$. Ahora no es posible resolverla integrando directamente (ya que no es posible despejar y' solamente en función de x), pero sí es posible transformar la EDO en una de variables separadas (haciendo las operaciones necesarias), de manera que se obtiene $dy/(1-y)=dx/x$, por lo que podremos integrar ambos miembros e igualarlos para obtener la solución buscada.

```
--> /*EJEMPLO N° 16:*/
```

```
    integrate(1/(1-y),y);
    integrate(1/x,x);
```

A continuación podemos escribir la solución igualando ambos resultados y añadiendo la constante de integración en uno de los miembros (la llamaremos "C"). Al usar solve() de la siguiente manera, Maxima despejará la función y si es posible:

```
--> /*EJEMPLO N° 16 (cont.):*/
```

```
    solve(integrate(1/(1-y),y)=integrate(1/x,x)+C,y);
```

Para terminar observemos que escribir $-e^{(-C)}$ es (casi) lo mismo que escribir C , puesto que se trata de una constante al fin y al cabo; pero el programa no hace esa simplificación por sí solo. En este caso funciona mejor la resolución directa con ode2:

```
--> /*EJEMPLO N° 16 (cont.):*/
```

```
    ode2(x*'diff(y,x)+y=1,y,x);
```

```
--> /*EJEMPLO N° 17: Para resolver  $y*y'+x=0$  lo haremos directamente con ode2:*/
```

```
    ode2(y*'diff(y,x)+x=0,y,x);
    solve(%,y);
```

EJERCICIO N° 3:

a) Comprueba que las soluciones que hemos obtenido en los anteriores ejemplos son correctas sustituyendo en la ecuación diferencial.

b) Resuelve mediante separación de variables la ecuación

$$x^2*y'-y=1$$

Resuélvela después usando ode2.

c) Resuelve la siguiente edo usando ode2:

$$y'=(3x^2+4x+2)/(2y-2)$$

Tomando entonces como constante el valor 3, representa las soluciones en $[-2,2]$ usando plot2d.

3.3 EDO's homogéneas y exactas

✓ También podemos usar el programa para comprobar si una ecuación es homogénea o exacta, aunque en este caso tendremos que darle a Maxima las operaciones que queremos que nos calcule. Por ejemplo, para ver si una EDO es homogénea, tendremos que escribirla en la forma $M(x,y)dx+N(x,y)dy=0$, y será HOMOGÉNEA si se verifica que $M(tx,ty)=t^n M(x,y)$ y $N(tx,ty)=t^n N(x,y)$. Por tanto, los dos miembros de estas igualdades son las que tendríamos que calcular usando Maxima (notemos que posiblemente sea más sencillo hacerlo a mano). De igual forma, sabemos que la EDO será EXACTA si se verifica que $@M/@y=@N/@x$, por lo que tendremos que pedirle a Maxima que nos calcule ambos miembros para ver si son iguales.

✓ EJERCICIO N° 4:
En las siguientes EDO'S decir si son homogéneas, exactas, ambas cosas o ninguna:

a) $(y^2+y*x)dx+(2*x*y+(x^2)/2)dy=0$. (Sol: exacta y homogénea)

b) $y/x+y'\log(x)=0$. (Sol: exacta y no homogénea)

c) $x*\exp(y/x)dx+ydy=0$. (Sol: no exacta y sí homogénea)

✓ Para resolver estas ecuaciones podemos seguir (con ayuda de Maxima) las indicaciones de la teoría (por ejemplo, para el caso de las EDO homogénea realizar el cambio $y=x*u(x)$, que nos da una edo en variables separadas x y u), aunque es más sencillo usar la indicación `ode2` (aunque normalmente nos dará siempre una solución en forma implícita)

✓ EJERCICIO N° 5:
Resolver, con ayuda de `ode2`, las tres EDO's del ejercicio anterior.

□ 3.4 Ecuaciones lineales de 1er orden

✓ En lugar de buscarle un factor integrante que dependa de x (que sabemos que siempre lo admiten) es más sencillo usar la fórmula vista en la teoría: Para una EDO lineal de 1er orden $y'+f(x)y=g(x)$ su solución viene dada por

$$y=\exp(-\text{integral}(f(x)))*(\text{Integral}(g(x)*\exp(\text{integral}(f(x))))+cte)$$

aunque estas ecuaciones también podemos resolverlas usando `ode2`:

✓

```
--> /*EJEMPLO N° 18: Podemos resolver y'-(sin(x))*y=sin(x)
      usando ode2:*/

      ode2('diff(y,x)-(sin(x))*y-sin(x)=0,y,x);
```

✓ EJERCICIO N° 6:
Resolver la EDO anterior usando la fórmula general del comentario anterior.

□ 3.5 Ecuaciones lineales de orden superior con coeficientes constantes

✓ Vamos a resolver EDO's lineales de orden superior y con coeficientes constantes. Veremos solamente las ecuaciones homogéneas.

Ejemplo 19. Resolver la ecuación homogénea
 $y^{(5)} - 7y^{(4)} + 14y''' + 14y'' - 47y' - 39y = 0$

Sabemos que se trata tan sólo de escribir el polinomio característico correspondiente a sustituir las derivadas de y por potencias de cierta variable ' m ' y hallar las raíces correspondientes:

En este caso tenemos el par de conjugadas $m=3+2i$ y las reales $m=3$ y $m=-1$, pero no está claro si estas raíces reales son simples o dobles. Para ello habrá que ir un poco más allá y pedir a Maxima que nos factorice el polinomio. Si lo hacemos, veremos que $m=-1$ es doble y $m=3$ es simple. Por lo tanto, la solución de la ecuación es
 $y = C1 * e^{(3*x)} + C2 * e^{(-x)} + C3 * x * e^{(-x)} + C4 * e^{(3*x)} * \cos(2*x) + C5 * e^{(3*x)} * \sin(2*x)$

Actualmente Maxima está muy limitado para resolver por sí sólo ecuaciones diferenciales de orden superior. Así que nos contentaremos con comprobar la solución obtenida.

✓ EJERCICIO N° 7:

a) Resuelve la ecuación homogénea $y''' - 3y'' + 2y' = 0$ y comprueba la solución.

b) Idem para $y''' - y'' - 4y' = 0$ y comprueba la solución.

□ 3.6 Sistemas de EDO's lineales

✓ La función 'desolve' resuelve sistemas de EDO's lineales utilizando la transformada de Laplace. La dependencia funcional respecto de una variable independiente debe indicarse explícitamente, tanto en las variables como en las derivadas. Por ejemplo, para la primera derivada, a diferencia de la forma 'diff(y,x) (usada en los apartados anteriores) debe usarse la forma 'diff(y(x),x).

✓ --> /*EJEMPLO N° 20: He aquí la solución de la ecuación diferencial
 $y''(x) + y(x) = 2*x$ */

```
desolve('diff(y(x),x,2)+y(x)=2*x,y(x));
```

✓ --> /*EJEMPLO N° 21: Para resolver el sistema $x'(t)=y(t)$; $y'(t)=x(t)$ */

```
desolve(['diff(x(t),t)=y(t),'diff(y(t),t)=x(t)],[x(t),y(t)]);
```

✓ --> /*EJEMPLO N° 22 (cont.): Podemos añadir condiciones iniciales al sistema anterior $x(0)=1$; $y(0)=0$ */

```
atvalue(x(t),t=0,1);
atvalue(y(t),t=0,0);
```

```
--> /*EJEMPLO N° 23 (cont.): Podemos volver a resolver el sistema con  
las condiciones iniciales dadas*/
```

```
desolve(['diff(x(t),t)=y(t)', 'diff(y(t),t)=x(t)'], [x(t), y(t)]);
```

```
--> /*EJEMPLO N° 24: He aquí la solución de la ecuación diferencial  
y'''(x)-3*y''+3*y'- y(x)=e^x*x^2 con las condiciones iniciales  
y(0)=1; y'(0)=0; y''(0)=-2*/
```

```
eq:'diff(y(x),x,3)-3*'diff(y(x),x,2)+3*'diff(y(x),x)-y(x)=%e^x*x^2;  
atvalue('diff(y(x),x,2),x=0,-2)$  
atvalue('diff(y(x),x),x=0,0)$  
atvalue(y(x),x=0,1)$  
desolve([eq],[y(x)]);
```